# Maintainer: Felix Yan <felixonmars@archlinux.org>

_hkgname=chell-quickcheck
pkgname=haskell-chell-quickcheck
pkgver=0.2.5.3
pkgrel=70
pkgdesc="QuickCheck support for the Chell testing library"
url="https://github.com/typeclasses/chell"
license=('MIT')
arch=('loong64' 'x86_64')
depends=('ghc-libs' 'haskell-chell' 'haskell-quickcheck' 'haskell-random')
makedepends=('ghc')
source=(https://hackage.haskell.org/packages/archive/${_hkgname}/${pkgver}/${_hkgname}-${pkgver}.tar.gz)
sha512sums=('d2f8eca8995c45aa3732303c7168f12d98b0b121ea5730dc481041a900b6c35b6ac965cc47eea6710f486c4ed850915daeacca1e8a8f844a242386c9dd854a96')

build() {
    cd $_hkgname-$pkgver
    
    runhaskell Setup configure -O --enable-shared --enable-executable-dynamic --disable-library-vanilla \
        --prefix=/usr --docdir=/usr/share/doc/$pkgname --datasubdir=$pkgname \
        --dynlibdir=/usr/lib --libsubdir=\$compiler/site-local/\$pkgid --ghc-option=-fllvm
    runhaskell Setup build $MAKEFLAGS
    runhaskell Setup register --gen-script
    runhaskell Setup unregister --gen-script
    sed -i -r -e "s|ghc-pkg.*update[^ ]* |&'--force' |" register.sh
    sed -i -r -e "s|ghc-pkg.*unregister[^ ]* |&'--force' |" unregister.sh
}

package() {
    cd $_hkgname-$pkgver
    
    install -D -m744 register.sh "$pkgdir"/usr/share/haskell/register/$pkgname.sh
    install -D -m744 unregister.sh "$pkgdir"/usr/share/haskell/unregister/$pkgname.sh
    runhaskell Setup copy --destdir="$pkgdir"
    install -D -m644 "license.txt" "${pkgdir}/usr/share/licenses/${pkgname}/license.txt"
    rm -f "${pkgdir}/usr/share/doc/${pkgname}/license.txt"
}
